##
# Quick Gestures
# Copyright (c) 2011 Tomi Paananen.
#

QT += core declarative

DEFINES += QG_LIBRARY
TEMPLATE = lib
TARGET = quickgesturesengine
VERSION = 0.2.0

MOC_DIR = ./moc

CONFIG(debug, debug|release) {
    # Enable debug printing if not pure release build.
    message(Debug printing enabled.)
    DEFINES += QG_DEBUG
}

# Uncomment the following to enable logging.
CONFIG += QG_LOG

include(./qgengine.pri)

warning(Building Quick Gestures engine as a library!)

symbian {
    message(Symbian build)
    TARGET.CAPABILITY = ReadUserData \
                        WriteUserData \
                        UserEnvironment
    TARGET.EPOCALLOWDLLDATA = 1
    SYMBIANBUILD_DEPENDENCYOFF = 1
    TARGET.UID2 = 0x1000008D
    TARGET.UID3 = 0xE6649B8D

    dynamiclibrary.sources = quickgesturesengine.dll
    dynamiclibrary.path = !:/sys/bin
    DEPLOYMENT += dynamiclibrary
}
else {
    DESTDIR = ../bin
}

unix:!symbian {
    maemo5 {
    }
    else {
        contains(DEFINES, DESKTOP) {
        }
        else {
            message(Harmattan build)
            DEFINES += Q_WS_MAEMO_6
            target.path = /usr/lib
        }
    }

    INSTALLS += target
}

# End of file.
